
Notice
SAS is a registered trademark or trademark 
of SAS Institute Inc. in the USA and other 
countries. ® indicates USA registration. 
Other product and brand names are 
trademarks or registered trademarks of their 
respective owners. 



To copy code from the slides
1. Make a second copy of the slides so that 
you always have a backup of the original.
2. Rename the second copy to change the 
file extension from pps to ppt.
3. Open the ppt version, and you will be 
able to copy any code, text, or image.



Structure of Presentation
• Why use the ODS TableEditor TagSet
• What I regard as significant distinguishing 
advantages of ODS tagsets over using the 
simpler solution of ODS HTML
• The Basics of ODS HTML
• Color-Coding without Red-Green
• Multiple WorkSheets per WorkBook
• WorkSheets with Multiple Graphs, Multiple 
Tables, Multiple Tables and One Graph
(NOTE: One Graph is the maximum, and it must 
be last if presented in a horizontal panel)
• Some Either/Or limitations of TableEditor



You Can Help . . .
First. If you have any solutions to SAS-to-Excel 
reporting, I would appreciate your sharing them.
If I use them for publication, I will give you 
conspicuous acknowledgement.
I’m not a user of Excel macros, VBA, etc.
SAS, ODS, DDE only.
Second. If you find any error in these slides, 
please let me know.
Thanks.



Scope of Presentation
• As much as preparation time permitted
• For a paper that covered ODS HTML, a bit on 
the ExcelXP tagset, and a bit on an earlier 
version of the MSOffice2K_x tagset (which could 
not do multi-sheet workbooks), see
http://www.mwsug.org/proceedings/2010/solutions/MWSUG-2010-167.pdf

• For more on those two tagsets, as well a good 
introduction to the TableEditor tagset, see these 
slides and my updated and expanded paper, to 
be written as part of a near future project. Send 
me an email if you want to receive the paper.



Not In Scope of Presentation
• How to enable SAS to control the format of 
numeric data, dates, etc. when passed to Excel 
will be covered in an update to these slides
• MSOffice2K IS NOT covered here,
but MSOffice2K_x, which has more capability,
IS covered here



Content of Presentation

• Picture of Output
• Code Used
• Code discussion limited
• All code will provided via email upon request

– to adapt
– for self-study
– to get clarified by reading at support.sas.com



What will be in Online Proceedings

• These slides
• DDE slides, paper, and toolkit



Paper, Code, Other References
• These slides are based on work in progress
• Formal long paper might be written this summer
• Code package for these slides
cannot be added to Proceedings immediately

• You can request via email:
• Code package
• List of references
• Bessler summaries, excerpts from references
• Updates as they become available
• Paper, if prepared later



Acknowledgement
• Some of the code here is derived from or related 
to or inspired by examples at support.sas.com or 
in papers by Chevell Parker.
• There are numerous papers about ExcelXP by 
Vincent DelGobbo. I’m not a fan of ExcelXP. It 
cannot handle graphs. So there is probably 
nothing here from his papers.
• The easiest way to find papers by any author on 
any subject related to SAS is by searching the 
index to the treasure trove of SAS lore
at www.lexjansen.com



Comment
I have done a lot of work of communication-
effective graphic design and construction, 
and written numerous papers. The graphs 
used here are simply for proof of concept 
(i.e., that they can be delivered). They are 
not meant as good examples of graphic 
design. I can provide a web index to my 
graphics papers, if desired.



Ways to Do SAS to Excel
with highly formatted output

• DDE (Dynamic Data Exchange)*
• ODS HTML
• ExcelXP tagset
• MSOffice2K tagset
• MSOffice2K_x tagset
• TableEditor tagset
*See conference proceedings for paper, slides, 
and toolkit of sample programs, macros, etc.



Why care about 
TagSets?

If you don’t use DDE,
there are things that

ODS HTML cannot do
(DDE can do more
than TagSets can)



TableEditor TagSet
IS

the only ODS solution
to create Pivot Tables



The Best Reason
to use

TableEditor TagSet





ods tagsets.tableeditor
file="%sysfunc(getoption(work))\temp.html"
options(
pivotrow="product_line"
pivotcol="quarter"
pivotdata="profit"
pivotpage="year"
auto_excel="yes"
excel_save_file="C:\\SAStoXLoutput\\PivotTable.xls"
);
proc print data=sashelp.orsales;
run;
ods tagsets.tableeditor close;
/* MORE pivot table options are available */



Another Thing
That

ONLY
TableEditor

Can Do



Excel-specific Format Table
and
Excel Graph in a Panel



SAS creates Excel Table & Excel Plot



ods tagsets.tableeditor 
file="%sysfunc(getoption(work))\temp.html"
options( 
excel_zoom="150"
/* other options here to create the Excel-only outputs */
auto_excel="yes"
excel_save_file=
"C:\\SAStoXLoutput\\Excel_Table_Excel_ScatterPlot.xls" 
);
proc print data=sashelp.class; run;
ods tagsets.tableeditor close;



options used to create the Excel-only outputs: 
sheet_name="Excel Table & Excel Plot"
chart_type="xyscatter"
chart_title="Excel Controls These Axis Ranges"
chart_title_size="10"
chart_title_color="8" /* this is disregarded */
chart_xaxes_title="Height"  chart_yaxes_title="Weight"
chart_area_color="6"  chart_plotarea_color="8"
chart_source="e:f" /* NOTE: SAS log contains a FALSE 
message:
"Unrecognized option: CHART_SOURCE" */
chart_location="same_sheet"
chart_legend="none" /* undocumented option—used 
here to remove useless legend */
auto_format="color2" /* Optional. Other values are ? */



Created at 1920 X 1080 and Cropped Right



Created at 1024 X 768 with zoom=“78”



Many, but not all uses of the TableEditor tagset,
require use of this type of program structure:
ods results; /* comments about this statement later */
ods tagsets.tableeditor
file="%sysfunc(getoption(work))\temp.html"
options( . . . 
auto_excel="yes"
excel_save_file="C:\\SAStoXLoutput\\FileName.xls");

< some PROC step here >
ods tagsets.tableeditor close;
%sysfunc(getoption(work)) assigns the folder used for 
your WORK data sets to be used for the HTML file.
The WORK folder is always deleted at end of session.
The path to the Save File (to your output Excel file)
must include \\ everywhere you would otherwise use \



ods tagsets.tableeditor
file="%sysfunc(getoption(work))\temp.html"
options( . . . 
auto_excel="yes"
excel_save_file="C:\\SAStoXLoutput\\FileName.xls");

< some PROC step here >
ods tagsets.tableeditor close;
ALWAYS causes two prompts. Here is the first:



As soon as you click Yes,
the Excel workbook is created,
but another prompt will appear . . .
If you use “ods noresults;”
instead of “ods results;” (which is the default),
NO prompts will appear
and NO Excel files will be created



Whether you click Yes or No,
the workbook is already complete.



Distinguishing Advantages
of ODS TagSets over ODS HTML



High Usability Features in Excel
- Titles
- Frozen Headers*
- Frozen RowHeaders*
- AutoFilters*
- Hyperlink in a title
- Zoom*
*Not available with ODS HTML,
Requires an ODS tagset



Effect of Each TagSet

- Enables each feature IF supported,
but possibly somewhat differently

- For the same ODS style
renders appearance of style differently

- In most cases,
I have used STYLE=Minimal



Support for AutoFilters

ExcelXP – All Columns or Selective, easy
MSOffice2Kx – Only All Columns option
Table Editor – All Columns or Selective, but
Different Syntax
Different Appearance of Filters



ExcelXP Frozen Hdrs & RowHdrs, Filter Cols A-D



Title1 Defect Using STYLE=Minimal



ods noresults; /* prevents pop-up about Save or Open */
ods graphics off; goptions reset=all;
ods tagsets.ExcelXP path="C:\SAStoXLoutput" 
style=minimal
file="SomeFileName.xml"  
/* can use file extension xls instead,
but it causes a misleading message at file open time */
options(

zoom="65"
embedded_titles="yes"
frozen_headers="5"  frozen_rowheaders="3"
autofilter="1-4"
sheet_name="ExcelXP Features Demo" );
< PROC PRINT STEP HERE >
ods tagsets.ExcelXP close;



/* PROC PRINT STEP: */
proc print data=sashelp.shoes noobs;
title1 height=16pt color=blue ' title1 text here ';
title2 height=16pt 'If using justify=left for TITLE1, what 
extends past Column A is hidden';
title3 justify=left height=16pt color=red 
link='http://www.wiilsu.org' 
'Go to Wisconsin Illinois SAS Users Home Page';
/* title3 "<a href='http://www.wiilsu.org'>Go to 

Wisconsin Illinois SAS Users Home Page</a>";
The above way to create a hyperlink in the title
does not work with this tagset. */

id region subsidiary product;
run;



Title1 Defect Removed If Using STYLE=Default



MSOffice2K_x Frozen Hdrs & RowHdrs, Filter All Cols



ods noresults; /* prevents pop-up about Save or Open */
ods graphics off; goptions reset=all;
ods tagsets.msoffice2k_x path="C:\SAStoXLoutput"
style=minimal
file="SomeFileName.xls"
options(

zoom="120"
frozen_headers="7"  frozen_rowheaders="3"
autofilter="yes"
sheet_name="MSOffice2K_x Features Demo" );
< PROC PRINT STEP HERE >
ods tagsets.msoffice2k_x close;



/* PROC PRINT STEP: */
proc print data=sashelp.shoes noobs;
title1 justify=left height=14pt color=blue 'title1 text';
title2 justify=left height=14pt color=red 'font color and 
height requested for link below is disregarded.';
/* title3 justify=left height=16pt color=red 

link='http://www.wiilsu.org' 'Go to Wisconsin Illinois SAS 
Users Home Page';
The above way to create a hyperlink in the title
does not work with this tagset. */

title3 justify=left height=14pt color=red "<a 
href='http://www.wiilsu.org'>Go to Wisconsin Illinois SAS 
Users Home Page</a>";
id region subsidiary product;
run;



TableEditor Frozen Hdrs Filter Cols A,B,C,D



ods results; /* MUST BE TURNED ON
so you can get and respond to prompts */

ods graphics off; goptions reset=all;
ods tagsets.tableeditor style=Minimal
file="%sysfunc(getoption(work))\temp.html"

options( sheet_name="TableEditor Features Demo 1"
excel_frozen_headers="yes"
autofilter="yes"
filter_cols="1,2,3,4" /* puts filters in row 2 */
/* AUTOFILTER_ENDCOL alternative does not work */
auto_excel="yes"
excel_save_file="C:\\SAStoXLoutput\\Demo1.xls" );
proc print data=sashelp.shoes noobs;
id region subsidiary product; run;
ods tagsets.tableeditor close;



TableEditor – AutoFilter all columns



ods results; /* MUST BE TURNED ON
so you can get and respond to prompts */

ods graphics off; goptions reset=all;
ods tagsets.tableeditor style=Minimal
file="%sysfunc(getoption(work))\temp.html"

options( sheet_name="TableEditor Features Demo 2"
excel_frozen_headers="yes"
autofilter="yes"
excel_autofilter="yes" /* All, no = None */
auto_excel="yes"
excel_save_file="C:\\SAStoXLoutput\\Demo1.xls" );
proc print data=sashelp.shoes noobs;
id region subsidiary product; run;
ods tagsets.tableeditor close;



ODS TagSets
• tagsets updated without new release of SAS
• part of Base SAS
• can be user installed*
• documentation ordeal

• too much PARTIAL information
• in TOO MANY places
• some of it wrong or out of date
• no manuals

• get doc in SAS log
by including doc=‘help’ in ODS statement*

*explained at support.sas.com,
I can provide explanation later.



Coping with the Documentation Ordeal
- these slides attempt to provide
a single, all-in-one-place survey
of all of the non-DDE ways
to create highly formatted reports in Excel
directly from SAS,
exclusively with SAS,
without getting involved with
Excel macros, VBA,
or other non-SAS machinery

- these slides are the first edition
of a work in progress



What I Can Do without DDE?
Solutions for Some Problems 

• Simple things, using ODS HTML
• Several examples
• Some are progressive learning exercises

• Also ODS tagsets
• ExcelXP, the most beloved of SAS Institute
• MSOffice2K_x*
• TableEditor, the ODS Pivot Table solution

• I still like DDE
*MUCH more powerful than MSOffice2K



What the heck is DDE?
Dynamic Data Exchange:
A concurrent Excel session
is started by the SAS session
and acts as a server
for that SAS session.
The SAS session acts as a client of Excel. 



What Can You Do with DDE?
• Anything you can do with keyboard & mouse
• If interested, see my DDE paper, slides, and 
toolkit in the Proceedings
Now let’s see what we can do without DDE



The Basics of ODS HTML



Getting Started





* Getting Started *;
* Non-Customized Excel Output *;
* KEY to the method: Use ods HTML to create XLS file *;

ods listing close;
ods noresults; * Do not open output in SAS *;

goptions reset=all; * Always do this. *;
ods HTML file="C:\SAStoXLoutput\GettingStarted.XLS";
* start your reporting code here *;
proc print data=sashelp.class noobs label;
run;
* end   your reporting code here *;
ods HTML close;



Using ODS Style Minimal





•Using ODS Style Minimal *;

goptions reset=all;
ods html
style=Minimal
file="C:\SAStoXLoutput\UsingODSstyleMinimal.xls";

proc print data=sashelp.class noobs label;
run;
ods html close;



Default Handling of Title





•Default Handling of Title *;

goptions reset=all;
ods html style=Minimal
file="C:\SAStoXLoutput\DefaultHandlingOfTitle.xls";

title1 'Title text (default or custom)';
title2 'always goes in column A';
title3 'and stretches it.';
proc print data=sashelp.class noobs;
run;
ods html close;



Using HTML Column Spanning Tag





•Using HTML Column Spanning Tag *;

goptions reset=all;
ods html style=Minimal
file="C:\SAStoXLoutput\TitleSpans4Columns.xls";

title1 
'<td COLSPAN=4>Title lines using HTML tag to span 

four columns</td>';
title2 . . . ;
title3 . . . ;
title4 . . . ;
proc print data=sashelp.class noobs;
run;
ods html close;



Title Spanning Seven Columns





* Title Spanning Seven Columns *;
* The Correct Solution *;
* for this length and font size of title *; 

goptions reset=all;
ods html style=Minimal
file="C:\SAStoXLoutput\TitleSpans7Columns.xls";

title1 
'<td COLSPAN=7>Title lines using HTML tag to span 

seven columns</td>';
title2 . . .; 
proc print data=sashelp.class noobs;
run;
ods html close;



Conventional TITLE Statement 
Controls for Appearance of Text 

Fail with Column Spanning





* Conventional TITLE Statement Controls for 
Appearance of Text Fail with Column Spanning *;

goptions reset=all;
ods html style=Minimal
file="C:\SAStoXLoutput\FailedTitleCustomization.xls";

title1 color=blue height=16pt font=Georgia
'<td COLSPAN=8>color=blue height=16pt
font=Georgia used for this TITLE statement</td>';

proc print data=sashelp.class noobs;
run;
ods html close;



Irrelevant Choice:
Customize Text Appearance

in Stretched Column A





* Irrelevant Choice: Customize Text Appearance
in Stretched Column A *;

goptions reset=all;
ods html style=Minimal
file="C:\SAStoXLoutput\IrrelevantChoice.xls";

title1 color=blue height=16pt font=Georgia
'color=blue height=16pt font=Georgia used for all 

TITLE text in Column A';
proc print data=sashelp.class noobs;
run;
ods html close;



Customized Column-Spanning Title





* Customized Column-Spanning Title *;

goptions reset=all;
ods html style=Minimal
file="C:\SAStoXLoutput\CustomizedColumnSpanning
Title.xls";

title1
'<td COLSPAN=8><font color=blue size=4
face=Georgia>HTML font controls used for this TITLE
statement</font></td>';

* HTML font sizes are 1,2,3,4,5,6,7,
which are not point sizes *;

proc print data=sashelp.class noobs;
run;
ods html close;



Table with Footnote





* Table with Footnote *;

goptions reset=all;
ods html style=Minimal
file="C:\SAStoXLoutput\TableWithFootnote.xls";

title1
'<td COLSPAN=8>Where do footnotes go in the
spreadsheet?</td>';

footnote1
'<td COLSPAN=8>I am the FOOTNOTE1 statement
text.</td>';

proc print data=sashelp.class noobs;
run;
ods html close;



Color-Coded Data



Color-Coded Data
Same Technique Applicable for

ODS HTML
ODS TagSets

ODS PDF
ODS RTF (for Word documents)

etc.





* Color-Coded Data *;

proc format lib=work;
value $SexCol
'F' = 'CXFF9999'
'M' = 'CX9999FF';
run;

< all usual ods html start stuff the same goes here >
proc print data=sashelp.class noobs;
var Name;
var Sex / style = [background = $SexCol.];
var Age Height Weight;
run;
ods html close;



PROC PRINT
Font and Color Control



PROC PRINT
Font and Color Control

Every part you can customize
except what’s accessible with 

STYLE(OBS) and 
STYLE(OBSHEADER)



PROC PRINT
Font and Color Control

With in-PROC STYLE controls
you can do MUCH MORE

than manage fonts and colors,
but that’s another presentation



PROC PRINT
Font and Color Control

The in-PROC STYLE controls
are useful even when your report 

destination is NOT Excel
(e.g., for PDF, RTF, or web)





* PROC PRINT Font and Color Control *;
proc sort data=sashelp.class out=ToPrint;
where name in ('Alice' 'Barbara' 'Carol' 'Judy');
by Age Name; run;

%let FontFormatting = font_face=Arial font_weight=Bold 
font_size=5;
%let TitleFormatting = face=Arial weight=Bold size=5 
color=black;

< familiar ODS HTML prefix code goes here >
< TITLE statement using column spanning and HTML 
font control >
< PROC PRINT step goes here >
ods html close;



About %LET

It stores text associated with a symbol in a table.

References to the symbol in your code get 
resolved at program run time (i.e., symbol is 
replaced by the text).

The text string stored for you begins with first 
non-blank after the = sign.

The text string stored for you ends with last non-
blank before the semi-colon.



About Symbols

& as prefix of symbol name is a signal

SAS replaces symbol at run time

If &YourSymbolName must be used in quotes,
you must use double quotes: “&MySymbolText”

At runtime ‘&MySymbolText’ gets resolved to
&MySymbolText, rather than to the desired text.



* Title Formatting *;

%let TitleFormatting = face=Arial weight=Bold size=5 
color=black;

title1 justify=left
"<td COLSPAN=6><font &TitleFormatting>PROC
PRINT Font & Color Control</font></td>"
justify=left
"<td COLSPAN=6><font &TitleFormatting>with ODS
STYLE=</font></td>";



* PROC PRINT Font and Color Control *;

proc print data=ToPrint label N / ... ;
by age;
id age / ... ;
var Name / ... ;
var Height;
sum Height / ... ;
sumby age;
run;

where ... are font and color controls



* PROC PRINT Font and Color Control *;

proc print data=ToPrint label N
style(header) = [&FontFormatting]
style(data)   = [&FontFormatting]
style(total)  = [&FontFormatting . . .]
style(bylabel) = [&FontFormatting . . .]
style(grand)  = [&FontFormatting . . .]
style(N)      = [&FontFormatting
background=cyan foreground=magenta just=left];



* PROC PRINT Font and Color Control *;

style(total)  = [&FontFormatting
background=magenta foreground=cyan]

style(bylabel) = [&FontFormatting
background=magenta foreground=cyan
just=center]

style(grand)  = [&FontFormatting
background=brown foreground=orange]



* PROC PRINT Font and Color Control *;

id age /
style(header) = [&FontFormatting
background=blue foreground=red]

style(data)   = [&FontFormatting
background=red foreground=blue];

var Name /
style(header) =
[background=black foreground=CX999999]

style(data)   =
[background=CX999999 foreground=black];



* PROC PRINT Font and Color Control *;

sum Height /
style(header) =
[background=CX009900 foreground=yellow]

style(data)   =
[background=yellow foreground=CX009900]

style(total)  =
[&FontFormatting
background=black foreground=white];



Composites
Not Shown,
but ODS HTML can do:
multiple tables per web page
multiple graphs per web page



Table Above Graph





* Table Above Graph *;

ods html style=Minimal path="C:\SAStoXLoutput"
file="TableAboveGraph.xls";

< proc print step here >
< proc gchart step here > 
ods html close;

* path= indicates where both the .xls file and the 
imbedded image file for the graph are stored *;



Table at Left of Graph
(and Sheet Name Customization)





* Table at Left of Graph *;
* (and Excel Sheet Name Customization) *;

ods tagsets.MSOffice2K_x style=Minimal
path="C:\SAStoXLoutput" 
file="TableAtLeftOfGraph.xls" 
options(panelcols="2"

sheet_name="Table At Left Of Graph" 
doc="help");

< proc print step here >
< proc gchart step here >
ods _all_ close;

* doc="help" to list all of the tagset options in SAS log *;



MSOffice2K_x panels

In this context,
panel = one row
of tables
or graphs
or tables and one graph



MSOffice2K_x panels
. graphs only – any number OK
. tables only – any number OK
. any number of tables

at left of one graph OK
. extra graphs (>1) right of table(s)

will overlay each other
. table at right of graph goes

below graph, not next to it



extra graphs (>1) right of table(s)
will overlay each other





table at right of graph goes
below graph, not next to it





Now . . .
Multi-Sheet Workbook
(using ExcelXP tagset,

which can not support graphs)

NOTE: MSOffice2K_x tagset
can do multi-sheet workbook

that includes graphs









* Multi-Sheet Workbook with ExcelXP TagSet *;
* Combination of
successive PROCs and BY processing *;

ods tagsets.excelxp style=Minimal file=" . . . ";
ods tagsets.excelxp options(sheet_name="Height");
proc print data=sashelp.class noobs;
id name; var height; run;
ods tagsets.excelxp options(sheet_name="none");
ods tagsets.excelxp options(sheet_interval='bygroup'
sheet_label='Sex' suppress_bylines='yes');

proc sort data=sashelp.class out=work.Sorted;
by sex; run;
proc print data=work.Sorted noobs;
by sex; id name; var age; run;
ods tagsets.excelxp close;



Multi-Sheet Workbook using 
MSOffice2K_x TagSet



Multi-Sheet Workbook using
MSOffice2K_x TagSet

Four Sheets:
- One Table (Simple Listing)
- Panel of three plots side by side
- Two Univariate Reports
Fifth Sheet (demo only) – A Warning



Let’s Open the Result



There is no corruption
Get this with every ODS tagset
if you create Excel output.
Exception: Create ExcelXP with 
filetype xml, but open it with Excel



This is a False Alarm
No Problem



This is a False Alarm
No Problem

But if you omit all graphs,
or create one per sheet
or multiple, but stacked within the sheet,
(i.e., avoid a side-by-side panel),
the pop-up does not occur.



For Demo Only

My workbook includes a warning worksheet











goptions reset=all;
ods graphics on / reset=all height=3in width=3in 
border=on antialiasmax=2500;
ods tagsets.msoffice2k_x path="C:\SAStoXLoutput" 
(url=none) file="Plots.html" style=minimal
options(orientation="landscape"
panelcols="3" panel_space="1" graph_width="225" );

ods graphics on / imagename='HgtVsAge'; 
proc sgplot data=sashelp.class;
title color=red " title text here ";
scatter y=height x=age; run; 
/* like above for two more plots */
ods tagsets.msoffice2k_x close;



ods tagsets.msoffice2k_x path="C:\SAStoXLoutput"
file="Table.html" style=minimal;

proc print data=sashelp.class noobs;
title justify=left height=12pt color=blue " title text here ";
var name age height weight;
run; 

ods tagsets.msoffice2k_x close;



ods noproctitle;

ods tagsets.msoffice2k_x path="C:\SAStoXLoutput"
file="HgtStats.html" style=minimal;

proc univariate data=sashelp.class;
title;
var height;
run;

ods tagsets.msoffice2k_x close; 

/* same for var weight */



ods tagsets.msoffice2k_x   path="C:\SAStoXLoutput"
file="MultiSheetWorkBook.xls"  style=minimal
options(worksheet_source=" 
Warning#.\WarningMessageFromLeRB.html,
OneTable#.\Table.html,
ThreePlots#.\Plots.html,
Hgt Stats#.\HgtStats.html,
Wgt Stats#.\WgtStats.html
") ;

data _null_;                                                                          
file print;                                                                            
put _all_;                                                                           

run;                                                                                      
ods tagsets.msoffice2k_x close;



About the dot . in OneTable#.\Table.html
dot is signal that html file is in the same 
folder as that of the output XLS file.

Package of xls file, html files,
and graph image files is relocatable
- Move anywhere
- Send via email
- XLS file will open & all parts will be found



Two Tables
in Separate WorkSheets
Using TableEditor TagSet





ods graphics off; goptions reset=all; ods results;
ods tagsets.tableeditor minimal
file="%sysfunc(getoption(work))\temp.html"

options(
sheet_name="Height,Weight“
excel_table_move="1,2"
auto_excel="yes"
excel_save_file=
"C:\\SAStoXLoutput\\TwoWorkSheets.xls"
);
proc print data=sashelp.class noobs;
var name age height; run;
proc print data=sashelp.class noobs;
var name age weight; run;
ods tagsets.tableeditor close;



Two Tables in a Panel
Using TableEditor TagSet





ods graphics off; goptions reset=all;
ods tagsets.tableeditor path="C:\SAStoXLoutput"
file="FileNameForTableEditorTwoTablePanel.xls"
style=minimal

options(
panelcols="2"
sheet_name="This Sheet Name Is Ignored"
);
title 'Height';
proc print data=sashelp.class noobs;
var name age height; run;
title 'Weight';
proc print data=sashelp.class noobs;
var name age weight; run;
ods tagsets.tableeditor close;



Table and Graph in a Panel
Using TableEditor TagSet



Table and Graph in a Panel
Using TableEditor TagSet

- Like MSOffice2K_x TagSet,
this only works if Table(s) is (are)
at left of Graph
- Trying to place Table at right of 
Graph pushes Table below it.
- Maximum of one graph without 
graph-graph overlay problem.





goptions reset=all
hsize=3.75in vsize=3.75in htext=4 PCT;

ods tagsets.tableeditor style=minimal
path="C:\\SAStoXLoutput" (url=none)
file="SAS_Table_and_SAS_ScatterPlot_SideBySide.xls"
options( panelcols="2" excel_zoom="75" );

/* zoom does not work */
proc print data=sashelp.class;
title1 "<td colspan=6><font color=red>SAS Table Using 
TableEditor TagSet</font></td>";
title2 "<td colspan=6><a href='http://www.wiilsu.org'>Get 
Info For Wisconsin Illinois SAS Users</a></td>"; run;
proc gplot data=sashelp.class;
title height=4 PCT "title text here";
plot weight*height / name='MyGPLOT'; run; quit;
ods tagsets.tableeditor close;



TableEditor TagSet

“Sometimes you can’t get
what you want”



The Torment of Either/Or



TableEditor TagSet

- TITLE statements are ignored
IF FrozenHeaders and/or AutoFilters 
are required

- In some cases, Excel_Zoom= is ignored
- In some cases, Sheet_Name= is ignored



Q: How can I create a zoomed worksheet with the 
TableEditor tagset?
A: You can not create the workbook directly, but 
must create it via an intermediate HTML file that 
gets programmatically exported to Excel using
EXCEL_FILE_SAVE



Q: How can I name a worksheet with the 
TableEditor tagset?
A: You can not create the workbook directly, but 
must create it via an intermediate HTML file that 
gets programmatically exported to Excel using
EXCEL_FILE_SAVE



ods noresults; /* Prevent pop-up asking you to Open, 
Save, or Cancel */
ods graphics off; goptions reset=all;
ods tagsets.tableeditor style=Minimal
path="C:\SAStoXLoutput"
file="DirectlyCreatedFileWithBothSheetNameSpecificati
onAndZoomRequestIgnored.xls"
options(
excel_zoom="50" /* This Option Is Ignored */
sheet_name="This Option Is Ignored, too"
);
title1 "title text here"; /* This is NOT ignored. */
proc print data=sashelp.shoes noobs;
run;
ods tagsets.tableeditor close;



Directly Created File



ods results; /* MUST BE TURNED ON
so you can get and respond to prompts */
ods graphics off; goptions reset=all;
ods tagsets.tableeditor style=Minimal
file="%sysfunc(getoption(work))\temp.html"
options(
excel_zoom="50"
sheet_name="WorkSheet with Zoom=50%"
auto_excel="yes"
excel_save_file="C:\\SAStoXLoutput\\WorksheetViaInter
mediateHTMLfileWithZoom50PercentAndWithRequeste
dSheetName.xls");
title1 'This title text does NOT appear';
title2 'even with option embedded_title="yes" ';
proc print data=sashelp.shoes noobs; run;
ods tagsets.tableeditor close;



SAS Code Saved Intermediate HTML To Excel



Q: How can I get a TITLE in my worksheet with 
the TableEditor tagset?
A: You must create the workbook directly,
NOT via an intermediate HTML file that gets 
programmatically exported to Excel using
EXCEL_FILE_SAVE



ods graphics off; goptions reset=all;
ods tagsets.tableeditor path="C:\SAStoXLoutput"
file="TitleMethod1.xls" style=Default;

/* Appearance of Default style can vary
based on SAS version,
but location of title does not.
Also, appearance of any style used
varies based on tagset used. */
title 'Listing of SASHELP.CLASS Data Set';
proc print data=sashelp.class noobs; run;
ods tagsets.tableeditor close;



Title in Column A of Row 1
Also, note absence of grid lines



ods graphics off; goptions reset=all;
ods tagsets.tableeditor path="C:\SAStoXLoutput"
file="TitleMethod2.xls"
style=Default;

title '<td colspan=7>Listing of . . . </td>';
/* Title (in Row 1) to span seven columns,
but ODS starts it in Column B */
proc print data=sashelp.class noobs;
run;
ods tagsets.tableeditor close;



Nice Custom-Formatted Column Spanning Title,
could br a hyperlink, but what about grid lines?



I want grid lines
in my worksheet created with the 
TableEditor tagset



Let’s try 
options(gridline=“yes” . . .)





Let’s try 
options(gridline_color=“red” . . .)





Q: How can I get gridlines in my worksheet with 
the TableEditor tagset?
A: You can not create the workbook directly, but 
must create it via an intermediate HTML file that 
gets programmatically exported to Excel using
EXCEL_FILE_SAVE



ods graphics off; goptions reset=all;
ods results;
ods tagsets.tableeditor style=Default
file="%sysfunc(getoption(work))\temp.html"

options(
sheet_name="Grid Lines Are Present"
auto_excel="yes"
excel_save_file="C:\\SAStoXLoutput\\GridLines.xls") ;
proc print data=sashelp.class noobs; run;
ods tagsets.tableeditor close;





Q: OK, now I know how to get gridlines with the 
TableEditor tagset. They require me to create the 
worksheet via an intermediate HTML file that gets 
programmatically exported to Excel using 
EXCEL_FILE_SAVE.
Suppose I must use this creation method for a 
different reason.
Can I TURN OFF the gridlines?
A. OK, let’s try.



ods graphics off; goptions reset=all;
ods results;
ods tagsets.tableeditor style=Default
file="%sysfunc(getoption(work))\temp.html"

options(
sheet_name="trying gridline=no"
gridline="no"
auto_excel="yes"
excel_save_file=
"C:\\SAStoXLoutput\\GridTurnOff_Attempt_1.xls") ;
proc print data=sashelp.class noobs; run;
ods tagsets.tableeditor close;





ods graphics off; goptions reset=all;
ods results;
ods tagsets.tableeditor style=Default
file="%sysfunc(getoption(work))\temp.html"

options(
sheet_name="trying gridline_color=red"
gridline_color="red"
auto_excel="yes"
excel_save_file=
"C:\\SAStoXLoutput\\GridTurnOff_Attempt_2.xls") ;
proc print data=sashelp.class noobs; run;
ods tagsets.tableeditor close;
/* NOTE: In the SAS Display Manager Results Viewer 
window, you can see the temp.html file,
for which THOSE grid lines are red. */



In the Results Viewer, with gridlinecolor=‘red’



In Excel, with gridlinecolor=‘red’



TableEditor and Grid Lines
Directly Created Files NEVER have grid lines.
Indirectly Created Files ALWAYS have grid lines,
and it is impossible to control their color.
The TableEditor TagSet options for grid lines
do NOT work as expected.
I will be pursuing this matter with SAS Technical 
Support. If I find a solution, I will update these 
slides.



Your Turn . . .
First. If you have any solutions to SAS-to-Excel 
reporting, I would appreciate your sharing them.
If I use them for publication, I will give you 
conspicuous acknowledgement.
I’m not a user of Excel macros, VBA, etc.
SAS, ODS, DDE only.
Second. If you find any error in these slides, 
please let me know.
Thanks.



Paper, Code, Other References
• These slides are based on work in progress
• Formal long paper might be written this summer
• Code package for these slides
cannot be added to Proceedings immediately

• You can request via email:
• Code package
• List of references
• Bessler summaries, excerpts from references
• Updates as they become available
• Paper, if prepared later



What will be in Online Proceedings

• These slides
• DDE slides, paper, and toolkit



Conclusions
Presentations and Papers are supposed to end 
with Conclusions.
It would have been nice to furnish a matrix of 
which tool to use for which task.
If you assemble one after you read these slides, 
and maybe do some investigation and 
experimentation of your own, kindly send it to me.
I explained that this is a work in progress.
You have seen where I have gotten so far.
I hope that you find the slides helpful. 



Coping with the Documentation Ordeal
- these slides attempt to provide
a single, all-in-one-place survey
of all of the non-DDE ways
(i.e., of all of the ODS ways)
to create highly formatted reports in Excel
directly from SAS,
exclusively with SAS,
without getting involved with
Excel macros, VBA,
or other non-SAS machinery

- these slides are the first edition
of a work in progress

